
We employ the dependently-typed programming language Agda2 to explore formalisation of untyped and
typed term graphs directly as set-based graph structures, via the gs-monoidal categories of Corradini and
Gadducci, and as nestedlet-expressions using Pouillard and Pottier’sNotSoFresh library of variable-binding
abstractions.

1 Introduction

The Coconut project [AK09a, AK09b] uses “code graphs” [KAC06], a variant of term graphs in the spirit
of “jungles” [HP91, CR93], as intermediate presentation for the generation of highly optimised assem-
bly code. This is currently implemented in Haskell, and we use the Haskell type system in an embedded
domain-specific language (EDSL) for creating such code graphs via what appears to be standard Haskell
function definitions, withlet-definitions introducing sharing, and with functions representing assembly-
level operations constructing hyperedges [AK09a]. However, since Haskell does not support full depen-
dent typing, the intermediate term graph datatype interface, supporting graph navigation, traversal, and
manipulation operations, cannot preserve the connection with the Haskell-level typing of the assembly
operations. Therefore, although EDSL-created code graphsarewell-typed by construction, as certified
by the type checker, this does not hold anymore for code graphs that are the result of internal operations.
Those internal operations either require separate proof that they preserve well-typedness, or they need to
perform run-time checks, at considerable run-time cost.

In addition, our code-graph-creation EDSL has a second “simulator” implementation, which turns
the EDSL expressions into Haskell functions that implementa “machine simulation”. Since the code
graph representation has lost its connection with the Haskell-level typing, it is “unintuitively hard” to use
the simulation machinery for code graphs that result from code graph manipulation operations.

Mainly for these reasons, we are now exploring implementation of code graphs in a dependently
typed programming language, where there is no need to “loose” the type information when moving to a
graph representation, and where even stronger assertions about operations on code graphs than just type
preservation can be proveninsidethe implementing system.

We start, in Sect. 2, with a quick introduction to the dependently typed programming language (and
proof checker) Agda [Nor07]. This is followed by formalisations of set-based mathematical definitions
of untyped (Sect. 3) and typed (Sect. 4) term graphs, and thena summary of the gs-monoidal category
view on these term graphs in Sect. 5. Finally, we present two formalisations of acyclic term graphs as
(differently structured) nestedlet-expressions (Sections 6 and 7).

2 Introduction to Agda: Types, Sets, Equality

The Agda home page1 states:

1http://wiki.portal.chalmers.se/agda/

http://dx.doi.org/10.4204/EPTCS.48.6
http://wiki.portal.chalmers.se/agda/

Agda is a dependently typed functional programming language. It has inductive families,
i.e., data types which depend on values, such as the type of vectors of a given length. It also has
parametrised modules, mixfix operators, Unicode characters, and an interactive Emacs interface
which can assist the programmer in writing the program.

Agda is a proof assistant.It is an interactive system for writing and checking proofs.Agda is
based on intuitionistic type theory, a foundational systemfor constructive mathematics developed
by the Swedish logician Per Martin-Löf. It has many similarities with other proof assistants based
on dependent types, such as Coq, Epigram, Matita and NuPRL.

Syntactically and “culturally”, Agda is quite close to Haskell. However, since Agda is strongly normal-
ising and has no⊥ values, the underlying semantics is quite different. Also,since Agda is dependently
typed, it does not have the distinction that Haskell has between terms, types, and kinds (the “types of the
types”). The Agda constantSet corresponds to the Haskell kind*; it is the type of all “normal” datatypes.
For example, the Agda standard library defines the typeBool as follows:

data Bool : Set where true : Bool
false : Bool

SinceSet needs again a type, there isSet1, with Set : Set1, etc., resulting in a hierarchy of “universes”.
Since version 2.2.8, Agda supportsuniverse polymorphism, with universesSet i wherei is an element of
the following special-purpose variant of the natural numbers:

data Level : Set where zero : Level
suc : (i : Level)→ Level

With this, the conventional usage turns into syntactic sugar, so thatSet is now Set zero, andSet1 =
Set (suc zero). For example, the standard library includes the following universe-polymorphic definition
for the parameterisedMaybe type:

data Maybe {a : Level} (A : Set a) : Set a where just : (x : A)→ Maybe A
nothing : Maybe A

Maybe has two parameters,a andA, where dependent typing is used since the type of the second param-
eter depends on the first parameter. The use of{...} flagsa as animplicit parameterthat can be elided
where its type is implied by the call site ofMaybe. This happens in the occurrences ofMaybe A in the
types of the data constructorsjust andnothing: In Maybe A, the value of the first, implicit parameter of
Maybe can only bea, the level of the setA.

The same applies to implicit function arguments, and in mostcases, implicit arguments or parameters
are determined by later arguments respectively parameters. Frequently, implicit arguments correspond
quite precisely to that part of the context of mathematical statements that is frequently left implicit by
mathematicians, so that the reader may be advised to skip implicit arguments at first reading of a type, and
return to them for clarification where necessary for understanding the types of the explicit parameters.

While the Hindley-Milner typing of Haskell and ML allows function definitions without declaration
of the function type, and type signatures without declaration of the universally quantified type variables,
in Agda, almost all types and variables need to be declared, but implicit parameters and the type checking
machinery used to resolve them alleviate that burden significantly. For example, the original definition
writes onlyMaybe {a} (A : Set a) : Set a, since the type ofa will be inferred froma’s use as argument
to Set.

The “programming types” likeMaybe can be freely mixed with “formula types”, inspired by the
Curry-Howard-correspondence of “formulae as types, proofs as terms”. The formula types of true for-
mulae contain their proofs, while the formula types of falseformulae are empty.

The standard library type of propositional equality has (besides two implicit parameters) one explicit
parameter and one explicit argument; the definition therefore gives rise to types like the type “2≡ 1 + 1”,
which can be shown to be inhabited using the definition of natural numbers1 and2 and natural number
addition+, and the type “2 ≡ 3”, which is an empty type, since it has no proof.

data _≡_ {a : Level} {A : Set a} (x : A) : A → Set a where refl : x ≡ x

The underscore characters occurring in the name_≡_ declare mixfix syntax with argument positions for
explicit parameters and arguments; this mixfix syntax is already used in the type of the single constructor.
The definition introduces typesx ≡ y for anyx andy of typeA, but only the typesx ≡ x are inhabited,
and they contain the single elementrefl {a} {A} {x}.

In Agda, as in other type theories without quotient types, sets with equality are typically modelled as
setoids, that is, carrier types equipped with an equivalence. This closely corresponds to the non-primitive
nature of the “equality” test(==) : Eq a ⇒ a → a → Bool in Haskell. A setoid is a dependent record
consisting of aCarrier set, a relation_≈_ on that carrier, and a proof that the relation_≈_ is an
equivalence relation:

record Setoid c l : Set (suc (c⊔l)) where

field Carrier : Set c
_≈_ : Rel Carrier l
isEquivalence : IsEquivalence _≈_

open IsEquivalence isEquivalence public

An Agda record is also a module that may contain other material besides itsfields; the “open” clause
makes the fields of the equivalence proof available as if theywere fields ofSetoid. This language feature
enables incremental extension of smaller theories to larger theories at very low notational cost.

Whenever we allow arbitrary node or edge sets, and we want to prove, for example, isomorphism of
certain graphs, we actually need setoids and not just sets. For such contexts, we introduce the following
abbreviation for extracting the carrier set from a setoid:

⌊_⌋ : {c l : Level}→ Setoid c l → Set c
⌊ s ⌋ = Setoid.Carrier s

3 Set-Based Term Graphs

We now present a simple definition of term graphs that is intentionally kept close to conventional math-
ematical formulations. To reduce complexity and improve readability of this initial formalisation, we
present untyped term graphs here; a typed variant will be shown in Sect. 4.

In the context of an arity-indexed label typeLabel : N→ Set, we first define a typeDHG1 of directed
hypergraphs with one putput per edge, indexed by input and output arities of the whole graph, with the
following components (since Agda records are also modules,they can contain additional material besides
theirfields):

• A setoidInner of non-input nodes. (For simplicity, we do not emply universe polymorphism here, and
all our setoids are of typeSetoid zero zero.)
For technical reasons, we find it more convenient to have the non-input nodes separate from the input
nodes. Otherwise we would have had to include an explicit injection from the input positions to the
complete node set.

• The setoidNode of all nodes is then derived as the disjoint union ofInner with the setoid of input
positions, which is obtained fromFin m, the set of natural numbers smaller thanm.

• The secondfield is then-element vector ofoutput nodes, which can be either input nodes or inner
nodes.

• For symmetry, we also provide them-element vector ofinput nodes, constructed usingallFin m which
is the vector (i.e., array) containing allm elements of the setFin m in sequence, i.e.,0, 1, . . . ,m - 1.

• Edge is the setoid of hyperedges.
• eInfo maps each edge to a dependent tuple consisting of an arityk, ak-ary label, and ak-element vector

of edge input nodes.
• eOut maps each edge to its output node, which cannot be an input node of theJungle, and therefore

has to be anInner node. (The function arrow between setoids is optically not distinguishable from
the general function type arrow, but is technically a different symbol. Since setoids cannot be used as
types, no confusion can arise.)

• We derive the functioneLabel that maps each edgee to its edge label. Since the arity of that label is
not known in advance, the functioneLabel returns a dependent pair consisting of the label arityk and
ak-ary label.

• We also derive the functioneIn that maps each edgee to the vector of input nodes ofe; the type of this
vector depends on the arity ofe, which is the first component (proj1) of the dependent tupleeLabel e.

record DHG1 (m n : N) : Set1 where

field Inner : Setoid zero zero
Node = Fin.setoid m ⊎⊎ Inner
field output : Vec ⌊ Node ⌋ n

input : Vec ⌊ Node ⌋ m
input = Vec.map inj1 (allFin m)

field Edge : Setoid zero zero
eInfo : ⌊ Edge ⌋

→ Σ [k : N ] (Label k × Vec ⌊ Node ⌋ k)

eOut : Edge → Inner

eLabel : ⌊ Edge ⌋ → Σ [k : N ] Label k
eLabel e = Product.map id proj1 (eInfo e)

eIn : (e : ⌊ Edge ⌋)→ Vec ⌊ Node ⌋ (proj1 (eLabel e))
eIn = proj2 ◦ proj2 ◦ eInfo

record Jungle (m n : N) : Set1 where

field Inner : Setoid zero zero
Node = Fin.setoid m ⊎⊎ Inner
field output : Vec ⌊ Node ⌋ n

input : Vec ⌊ Node ⌋ m
input = Vec.map inj1 (allFin m)

field Edge : Setoid zero zero
eInfo : ⌊ Edge ⌋

→ Σ [k : N] (Label k × Vec ⌊ Node ⌋ k)
EOut : Inverse Edge Inner

eOut : Edge→ Inner
eOut = Inverse.to EOut
producer : Inner → Edge
producer = Inverse.from EOut

eLabel : ⌊ Edge ⌋ → Σ [k : N ] Label k
eLabel e = Product.map id proj1 (eInfo e)

eIn : (e : ⌊ Edge ⌋)→ Vec ⌊ Node ⌋ (proj1 (eLabel e))
eIn = proj2 ◦ proj2 ◦ eInfo

In this DHG1 definition,eOut does not have to be surjective, which means that there may be “undefined
nodes”, andeOut also does not have to be injective, which means that there maybe “join nodes” in the
sense of [KAC06]. If bijectivity ofeOut is desired, we can replace the setoid mapping with an inverse
pair of mappings, and extracteOut and theproducer mapping for inner nodes from that, as shown above
to the right.

These jungles are isomorphic to conventional termgraphs, where inputs (as arguments) and labels are
attached directly to inner nodes:

record TermGraph (m n : N) : Set1 where

field Inner : Setoid zero zero
Node = Fin.setoid m ⊎⊎ Inner
field output : Vec ⌊ Node ⌋ n
input : Vec ⌊ Node ⌋ m
input = Vec.map inj1 (allFin m)
field label : ⌊ Inner ⌋ → Σ [k : N ] Label k

args : (n : ⌊ Inner ⌋)→ Vec ⌊ Node ⌋ (proj1 (label n))

The following basic constructor functions are highly similar forDHG1, Jungle, andTermGraph; we show
them here forJungle.

Using the one-element setoid⊤ (with elementtt), we can define primitive jungles consisting of a
single hyperedge:

prim : {k : N}→ Label k → Jungle k 1
prim {k} f = record

{Inner = ⊤
; output = [ inj2 tt ]
; Edge = ⊤
; eInfo = λ → (k,(f,Vec.map inj1 (allFin k)))
; EOut = Inverse.id
}

For wiring graphs, we need empty sets (⊥⊥) of edges and inner nodes:

wire : {m n : N}→ Vec (Fin m) n → Jungle m n
wire {m} {n} v = record

{Inner = ⊥⊥
; output = Vec.map inj1 v
; Edge = ⊥⊥
; eInfo = E.⊥⊥-elim
; EOut = Inverse.id
}

With this, we can easily construct the standard wiring graphs required for defining a gs-monoidal cate-
gory (see Sect. 5) ofJungles:

idJungle : {m : N}→ Jungle m m
idJungle = wire (allFin )

dupJungle : {m : N}→ Jungle m (m + m)
dupJungle {m} = wire (allFin m + allFin m)

termJungle : {m : N}→ Jungle m 0
termJungle = wire [ ]

exchJungle : (m n : N)→ Jungle (m + n) (n + m)
exchJungle m n = wire (Vec.map (raise m) (allFin n) + Vec.map (inject+ n) (allFin m))

Separating the inner nodes from the inputs in particular hasthe advantage that for sequential composition,
we can just use the disjoint union of the twoInner node sets:

seqJungle : {k m n : N}→ Jungle k m → Jungle m n → Jungle k n
seqJungle {k} {m} {n} g1 g2 = let

open Jungle
h1 : ⌊ Node g1 ⌋ → Fin k ⊎ (⌊ Inner g1 ⌋ ⊎ ⌊ Inner g2 ⌋)
h1 = Sum.map id inj1
h2 : ⌊ Node g2 ⌋ → Fin k ⊎ (⌊ Inner g1 ⌋ ⊎ ⌊ Inner g2 ⌋)
h2 = [(λ i → h1 (Vec.lookup i (output g1))), inj2 ◦ inj2 ]

′

in record

{Inner = Inner g1 ⊎⊎ Inner g2

; output = Vec.map h2 (output g2)
; Edge = Edge g1 ⊎⊎ Edge g2

; eInfo = [productMap22 (Vec.map h1) ◦ eInfo g1,productMap22 (Vec.map h2) ◦ eInfo g2]
′

; EOut = EOut g1 ⊕⊕ EOut g2

}

Parallel composition works similarly; here the input positions need to be adapted.

parJungle : {m1 n1 m2 n2 : N}→ Jungle m1 n1 → Jungle m2 n2 → Jungle (m1 + m2) (n1 + n2)
parJungle {m1} {n1} {m2} {n2} g1 g2 = let

open Jungle
h1 : ⌊ Node g1 ⌋ → Fin (m1 + m2) ⊎ (⌊ Inner g1 ⌋ ⊎ ⌊ Inner g2 ⌋)
h1 = Sum.map (inject+ m2) inj1
h2 : ⌊ Node g2 ⌋ → Fin (m1 + m2) ⊎ (⌊ Inner g1 ⌋ ⊎ ⌊ Inner g2 ⌋)
h2 = Sum.map (raise m1) inj2

in record

{Inner = Inner g1 ⊎⊎ Inner g2

; output = Vec.map h1 (output g1)
+ Vec.map h2 (output g2)

; Edge = Edge g1 ⊎⊎ Edge g2

; eInfo = [productMap22 (Vec.map h1) ◦ eInfo g1,productMap22 (Vec.map h2) ◦ eInfo g2]
′

; EOut = EOut g1 ⊕⊕ EOut g2

}

4 Typed Code Graphs

Coconut code graphs [KAC06] have types associated with nodes, and hyperedges may have not only
multiple inputs, but also multiple outputs, to be able to model operations that yield multiple results; the
typing of the input and output nodes needs to be compatible with the operations indicated by the edge
labels.

For simplicity, we assume here a global setType : Set of node types, and dispense with using
setoids in this section. An edge label is now indexed by vectors of input and output types, so we assume
Label : {m n : N} → Vec Type m → Vec Type n → Set, and also define the dependent record type
EdgeType for collecting these indices:

record EdgeType : Set where

field inArity : N
outArity : N
inTypes : Vec Type inArity
outTypes : Vec Type outArity

An edge label then is such an index collection together with alabel drawn from the corresponding label
set; theopen declaration makes theEdgeType fields available forEdgeLabel elements as if this was a
record extension:

record EdgeLabel : Set where

field eType : EdgeType
label : Label (EdgeType.inTypes eType) (EdgeType.outTypes eType)

open EdgeType eType public

For typed term graphs, there are many different ways to deal with node typing, and for any given way,
different views are useful in different contexts. We will keep a node typing function as afield, and derive
from this an indexed view of typed nodes, using the followinggeneral construct: Given a setA and a
typing functiontype for A, theType-indexed setTyped A type associates with every typety all elements
of A that have typety; formally, an element ofTyped A type ty is a dependent pair consisting of an
elementa : A together with a proof thattype a ≡ ty:

Typed : (A : Set)→ (A → Type)→ Type → Set
Typed A type ty = Σ [a : A ] (type a ≡ ty)

Since the Agda standard library does not provide a variant ofVec where the element types may depend
on their positions, we directly use dependently typed functions starting from these positions instead,
producing “typed vectors” with elements type according to the argument type vectorv:

TypedVec : (A : Set)→ (A → Type)→{k : N}→ Vec Type k → Set
TypedVec A type {k} v = (i : Fin k)→ Typed A type (Vec.lookup i v)

The EdgeInfo associated with each hyperedge then contains, besides anEdgeLabel, two such “typed
node vectors”, typed according to the label’s typing information (for modularity, this definition is kept
outside the code graph definition and parameterised with thetypeNodes for “typed node vectors” to be
supplied there):

record EdgeInfo (Nodes : {k : N}→ Vec Type k → Set) : Set where

field eLab : EdgeLabel
eInput : Nodes (EdgeLabel.inTypes eLab)
eOutput : Nodes (EdgeLabel.outTypes eLab)

open EdgeLabel eLab public

A CodeGraph is now defined roughly analogous to aJungle, with the following differences worth point-
ing out:
• Code graphs can be considered as “generalised hyperedges”,and therefore have anEdgeType derived

from theCodeGraph type parameters. Keeping the current parameters eases the implementation of the
categorical view, in comparison with using theEdgeType as a parameter instead.

• We only need to explicitly represent the typing of the inner nodes; from this we can derive the typing
of all Nodes by looking up the typing of the input positions ininTypes.

• A TypedNode ty is aNode with typety; an element ofTypedNodes v is a “typed node vector” accord-
ing to the type vectorv.

• TheCodeGraph field output and each individual edge interface useTypedNode “vectors”.
• We can still provide lower-level interfaces to edges; we show functions that extract the edge label, edge

input arity, and edge inputNode vectors (discarding the type information), both dependently-typed and
existentially-typed with respect to the vector length. (The corresponding functionseOut etc.are not
shown.)

record CodeGraph {m n : N} (inTypes : Vec Type m) (outTypes : Vec Type n) : Set1 where

cgType : EdgeType
cgType = record {inArity = m

; outArity = n
; inTypes = inTypes
; outTypes = outTypes}

field Inner : Set
iType : Inner → Type

Node = Fin m ⊎ Inner

nType : Node → Type
nType = [(λ i → Vec.lookup i inTypes), iType ] ′

TypedNode : Type → Set
TypedNode = Typed Node nType

TypedNodes : {k : N}→ Vec Type k → Set
TypedNodes = TypedVec Node nType

field output : TypedNodes outTypes

input : TypedNodes inTypes
input = λ i → (inj1 i, refl)

field Edge : Set
eInfo : Edge→ EdgeInfo TypedNodes

eLabel : Edge→ EdgeLabel
eLabel = EdgeInfo.eLab ◦ eInfo

eInArity : Edge→N

eInArity = EdgeInfo.inArity ◦ eInfo

eIn : (e : Edge)→ Vec Node (eInArity e)
eIn e = mkVec (proj1 ◦ EdgeInfo.eInput (eInfo e))

eIn′ : Edge→ Σ [k : N ] (Vec Node k)
eIn′ e = eInArity e,eIn e

Again, eOut is not guaranteed to reach all nodes, and, due to the possibility of multi-output operations,
this cannot be amended by joining theInner andEdge sets as in jungles. This and other degrees of
generality contained in this definition can be useful for certain purposes, but also can be forbidden for
other purposes by adding appropriate constraints.

We show the function for producing primitive one-edge code graphs:

prim : (l : EdgeLabel)→ CodeGraph (EdgeLabel.inTypes l) (EdgeLabel.outTypes l)
prim l = record

{Inner = Fin (EdgeLabel.outArity l)
; output = λ i → (inj2 i, refl)
; Edge = ⊤
; eInfo = λ → record {eLab = l

; eInput = λ i → (inj1 i, refl)
; eOutput = λ i → (inj2 i, refl)}}

While type-checking the three propositional equality proofs refl in here, Agda actually proves that the
mentioned types are indeed equal: An Agda program can only produceCodeGraph values that are cor-
rectly typed, both on the external interface, and internally at each port of each edge.



46 Dependently-Typed Formalisation of Typed Term Graphs

5 GS-Monoidal Categories

Corradini and Gadducci proposedgs-monoidal categoriesfor modelling acyclic term graphs [CG99];
extended discussion of how code graphs fit into this framework is contained in [KAC06]. Here we only
present a quick summary, and tie this into the formalisationin Sect. 3.

In a category theory context, we write “f : A →B” to declare that morphismf goes from objectA
to objectB, and use “.,” as the associative binarycompositionoperator; composition of two morphisms
f : A →B andg : B′→C is defined iffB = B′, and then(f ., g) : A →C . Furthermore, the identity
morphism for objectA is writtenIA .

Jungle can be seen to define morphisms of an untyped term graph category where objects are natural
numbers. (ForCodeGraph, the collection of Objects isΣ [k : N ] (Vec Type k).)

In theJungle category, a morphism fromm to n is an element ofJungle m n, that is, a term graph with
m input nodes andn output nodes. More precisely, such a morphism is an isomorphism class of jungles,
since node and edge identities do not matter; we will define aSetoid where theCarrier is Jungle m n and
equivalence proofs areJungle isomorphisms.

CompositionF .,G “glues” together the output nodes ofF with the respective input nodes ofG, as we
have implemented inseqJungle. The identity onn consists only ofn input nodes which are also, in the
same sequence, output nodes, and no edges, and is therefore constructed as a wiring graph:

idJungle : {m : N}→ Jungle m m
idJungle = wire (allFin )

Definition 5.1 A symmetric strict monoidal category[ML71] consists of a categoryC0, a strictly asso-
ciative monoidal bifunctor⊗ with 1l as its strict unit, and a transformationX that associates with every
two objectsA andB an arrowXA ,B : A ⊗B→B⊗A with:

(F⊗G) .,XC ,D = XA ,B
., (G⊗F) , XA ,B

.,XB,A = IA ⊗ IB ,

XA ⊗B,C = (IA ⊗XB,C )
., (XA ,C ⊗ IB) , X1l,1l = I1l .

For Jungle, the unit object 1l is the natural number0, and⊗ on objects is addition. On morphisms,
⊗ forms the disjoint union of code graphs, concatenating the input and output node sequences, as im-
plemented inparJungle. Xm,n differs from Im+n only in the fact that the two parts of the output node
sequence are swapped:

exchJungle : (m n : N)→ Jungle (m + n) (n + m)
exchJungle m n = wire (Vec.map (raise m) (allFin n) + Vec.map (inject+ n) (allFin m))

Definition 5.2 A strict gs-monoidal categoryis a symmetric strict monoidal category where in addition
! associates with every objectA of C0 an arrow !A : A →1l, and∇ associates with every objectA of
C0 an arrow∇A : A →A ⊗A , such thatI1l =!1l = ∇1l, and the following axioms hold:

∇A
., (IA ⊗∇A ) = ∇A

., (∇A ⊗ IA ) ∇A
.,XA ,A = ∇A ∇A

., (IA ⊗!A ) = IA

∇A ⊗B
., (IA ⊗XB,A ⊗ IB) = ∇A ⊗∇B !A ⊗B =!A ⊗!B

In Jungle, the “terminator” !n differs fromIn only in the fact that the output node sequence is empty.

termJungle : {n : N}→ Jungle n 0
termJungle = wire [ ]



Wolfram Kahl 47

The “g” of “gs-monoidal” stands for “garbage”: all edges of aterm graphG : m→n are garbage in the
term graphG.,!n.

The duplicator∇n in Jungle differs from In only in the fact that the output node sequence is the
concatenation of the input node sequence with itself:

dupJungle : {n : N}→ Jungle n (n + n)
dupJungle {n} = wire (allFin n + allFin n)

The “s” of “gs-monoidal” stands for “sharing”: every input of ∇k
., (F ⊗G) is shared byF : k→m and

G : k→n.
Code graphs (and term graphs) over a fixed edge label set form ags-monoidal category, but not a

Cartesiancategory, where in addition ! and∇ arenatural transformations, i.e., for allF : A →B we
haveF.,!B =!A andF ., ∇B = ∇A

., (F⊗F). To see how these naturality conditions are violated by term
graphs, the following fiveJungles correspond to the expressions below them (we draw jungles and code
graphs from the inputs on top to the outputs at the bottom, with numbered triangles marking input and
output positions, and rectangles enclosing edge labels).

F

0

0 0

F

0

F

0 1

0

F F

0 1

0

F : 1→1 !1 F ., !1 F ., ∇1 ∇1
., (F⊗F)

Formalising (symmetric gs-) monoidal categories in Agda isa straight-forward extension of the standard
type-theoretic formalisation of category theory derivingessentially from Kanda’s “effective categories”
[Kan81]; this uses setoids of morphisms, but not of objects.This approach is also used by Huet and
Saïbi [HS98, HS00] for their formalisation of category theory in Coq, and by Gonzalía [Gon06] for his
formalisation of Freyd and Scedrov’s allegory hierarchy [FS90] in Alf, a predecessor of Agda.

This approach also corresponds to the general practice in category theory to consider objects only up
to isomorphism, not up to equality. However, the definition of strict monoidal categories runs counter
to this approach, by assuming an object-level operation(⊗) satisfying non-trivial object-level equations.
Therefore we directly formalise what MacLane calls “relaxed” monoidal categories, with natural iso-
morphismsα : A ⊗ (B⊗C )→(A ⊗B)⊗C andλ : 1l⊗A →A andρ : A ⊗1l→A .

This explicit approach also has advantages for moving between different levels of data nesting with-
out requiring additional features; this is important for example for reasoning about the effect of SIMD
operations together with SIMD vector manipulations on individual scalar values, which is necessary for
verifying numerous high-performance “tricks”, see e.g.[AK08].

6 Term Graphs asLet Constructs

The code graph representation of Sect. 4 essentially is a typed variant of the current internal repre-
sentation of Coconut code graphs, but, as mentioned in the introduction, we essentially write Haskell
definitions to initially create code graphs.

In lazy functional programming implemented by graph reduction, since at least KRC [Tur82], local
definitions (vialet or where) are understood to introducesharing. In a mathematical context, [AK94]
represents cyclic term graphs as systems of mutually recursive equations, and [MOW98] presents sharing
in the call-by-needλ -calculus vialet-expressions.

In the following, we present two formalisations of term graphs defined by non-recursive nestedlet-
expressions. For the sake of readability, we restrict ourselves to untyped term graphs and single-output
primitives.

With let-expressions, we automatically have to deal with the complications of bound variables, in-
volving scoping, renaming to avoid variable clashes, etc. The Agda libraryNotSoFresh by Pouillard and
Pottier [PP10] allows us to abstract from these concerns to alarge degree, at the cost of following the
discipline of theirWorld-based programming interface. At the core of their approach, there areWorlds
in which different variables are in scope; for a worldα , the set of usable names isName α . Introducing
a new name happens via a “world extension link”; an element ofα ↼ β is aweak linkthat provides a
variable inβ that might be shadowing one of the variables inα , while an element ofα ↼→ β is astrong
link that provides, inβ , a variable that isfreshwith respect to all variables inα .

For programming and in mathematics, we are used to working ina context of weak links, while
symbol manipulation systems, including theorem provers and compilers, frequently disambiguate names
so that they can work with strong links exclusively. To enable both settings, we will parameterise over
these “worldExtension relation” with a parameterE : World → World → Set.

We first present the typeTG that formaliseslet-expressions with arbitrary nesting; this type is only
a slight modification of theλ -term datatypeTm from [PP10].

A value of typeTG E α m n is, in the context ofm input nodes and of a worldα providing already
existing inner nodes, a term graph “suffix” producingn output nodes:
• The input node at positioni can be produced as an output node byInput i.
• An existing nodex : Name α is produced as an output node byV x.
• The empty suffix is calledε .
• Given two suffixest andu of output lengthsn1 andn2, their union, with concatenated output lists, is

t▽ u. The symbol▽ reads “fork”, as in the fork algebras of [HFBV97]; it is related with the duplicator
∇ via the equationt ▽ u = ∇m

., (t⊗u).
• A primitive f can only be invoked while applying it to the outputs of a term graph suffixt and while

at the same time creating a new nodex in an expression of the shapeLet x f t u, which, in more
conventional notation, would read “let x = f (t) in u”.
If the primitive f expectsk inputs, the argument term graph suffixt, which may not use the new name
x because it is in the “old” worldα , has to havek outputs.
The term graph suffixu may use also the new namex, and its outputs will be the outputs of the
“Let x f t u” expression.

data TG (E : World → World → Set) (α : World) (m : N) : N→ Set where

Input : (i : Fin m) → TG E α m 1
V : (x : Name α)→ TG E α m 1
ε : TG E α m 0
_▽_ : {n1 n2 : N}→ TG E α m n1 → TG E α m n2 → TG E α m (n1 + n2)
Let : {β : World} {k n : N}

→ (x : E α β) -- let x
→ (f : Label k)→ (t : TG E α m k) -- = f (t)
→ (u : TG E β m n) -- in u
→ TG E α m n

Without additional support, defining term graphs using thisinterface is somewhat inconvenient — the
following assumes a unary labelF, a binary labelG, and a ternary labelH:

TG0 : Label 1 → Label 2 → Label 3 → TG _↼_ ø 3 1
TG0 F G H = let f0 = freshø -- a strong link

x0 = FreshPack.weakOf f0 -- weak view off0
n0 = FreshPack.nameOf f0 -- Name of f0

in Let x0 H
(Let x0 F (Input zero) (V n0 ▽ V n0)
▽

Let x0 G (Input (suc zero) ▽ Input (suc (suc zero))) (V n0)
)
(V n0)

F G

H

0

0 1 2

Using slightly more conventional notation, this corresponds to the following, relatively readable version,
with “ i” prefixing inputs and “n” prefixing node names:

let n0 = H ((let n0 = F (i0) in (n0 ▽ n0))
▽

(let n0 = G (i1 ▽ i2) in n0)
) in n0

Either by adding more notational support, or by defining a separate input language, this can provide an
interface that comes reasonably close to Haskell-style programming.

The real point of the definition ofTG however is that it not only provides an input language, but also
a representation of term graphs that can be manipulated and transformed by programs. For example, we
can turn aTG with name shadowing (i.e., using weak links) into one with strong links by replacing all
node names with fresh names relative to their respective worlds:

strengthenTG : {α α ′ : World}→ Fresh α ′ → CEnv (Name α ′) α
→{m n : N}→ TG _↼_ α m n → TG _↼→_ α ′ m n

strengthenTG ε = ε
strengthenTG fr Γ (t ▽ u) = (strengthenTG fr Γ t) ▽ (strengthenTG fr Γ u)
strengthenTG (Input i) = Input i
strengthenTG fr Γ (V x) = V (lookupCEnv Γ x)
strengthenTG fr Γ (Let x f t u)

= let Γ′ = mapCEnv importWith Γ,x 7→ nameOf
in Let strongOf f (strengthenTG fr Γ t) (strengthenTG nextOf Γ′ u)

where open FreshPack fr

Parallel composition is also easy to program, using fork after embedding, respectively shifting, the inputs:

parTG : {E : } {α : } {m1 n1 m2 n2 : N}
→ TG E α m1 n1 → TG E α m2 n2 → TG E α (m1 + m2) (n1 + n2)

parTG {E} {α} {m1} {n1} g1 {m2} {n2} g2 = extendTG m2 g1 ▽ shiftTG m1 g2

Sequential composition is much harder to implement directly, since the output nodes of the first argument
may have been defined in separate worlds and combined with fork, and now need to be brought into a
common world, which in general requires renaming and restructuring. A convenient “canonical form”
for suchlet-expressions has noLet at argument positions, and noLet below fork, and therefore degen-
erates into a sequence ofLet declarations each binding a new node to the application of some primitive

to existing nodes. When dealing with any kind of canonical forms, especially in a dependently-typed
setting, it is frequently worth while declaring this as a separate datatype so that it becomes easier to
exploit its properties. For this canonical form ofTG, we introduce a separate datatype with additional
restructuring below.

7 Term Graphs with Sequential Node Declaration

According to our explanation ofTG term graphs,▽ with ε obviously forms a monoid, but the monoid
laws do not come for free inTG. Moving to theVec container type instead provides us with the monoid
laws in the standard library, and makes for a more canonical representation. With this change, and with
strictly linearised node declaration, the term graphTG0 shown above could be written in a somewhat
conventional notation as follows (without fully specifying the number of inputs):

let n0 = F i0
let n1 = G i1 i2
let n2 = H n0 n0 n1
in [n2 ]

We introduce the typeArg for individual nodes, either existing inner nodes, or inputpositions, and a type
synonymArgs for their vectors:

data Arg α (m : N) : Set where

Input : (i : Fin m)→ Arg α m
V : (x : Name α)→ Arg α m

Args α m n = Vec (Arg α m) n

The datatypeTG′ has the same reading asTG, but a simpler structure:

• If all nodes have been declared,Output as assembles the vector of output nodes.
• Let x f v u, which, in more conventional notation, would read “let x = f (v) in u”, binds a new nodex

to an edge labelledf with input nodesv, and makesx visible in the remaining term graph suffixu.

data TG′ E α (m : N) : N→ Set where

Output : {n : N}→ Args α m n → TG′ E α m n
Let : {β : World} {k n : N}

→ (x : E α β ) -- let x
→ (f : Label k) (v : Args α m k) -- = f (v)
→ (u : TG′ E β m n) -- in u
→ TG′ E α m n

We first show that primitive and wiring graphs are easily programmed:

prim : {k : N}→ Label k → TG′ _↼→_ ø k 1
prim {k} f = Let strongOf f (Vec.map Input (Vec.allFin k)) (Output [V nameOf])

where open FreshPack freshø

wire : {k n : N} {E : } {α : World}→ Vec (Fin k) n → TG′ E α k n
wire v = Output (Vec.map Input v)

idWire : {k : N} {E : } {α : World}→ TG′ E α k k
idWire {k} = wire (Vec.allFin k)

dup : {k : N} {E : } {α : World}→ TG′ E α k (k + k)

dup {k} = wire (Vec.allFin k + Vec.allFin k)

term : {k : N} {E : } {α : World}→ TG′ E α k 0
term = wire [ ]

With these definitions, we can reconstruct the term graphTG0 from above via the gs-monoidal interface,
with sequential compositionseqTG′ and parallel compositionparTG′ defined below:

tg0 = seqTG′ (parTG′ (seqTG′ (prim F) dup) (prim G)) (prim H)

For the analogous function tostrengthenTG, which replaces each linkx in a Let construct with a fresh
link, we present an easy generalisation to serve dual purposes:
• Starting from weak links,strengthenTG′ {_↼_} id is proper strengthening;
• starting from strong links,strengthenTG′ {_↼→_} StrongPack.weakOf is renaming with fresh

names with respect to the new worldα ′.

strengthenTG′ : {E : }→ (E ⇒ _↼_)
→{α α ′ : World}→ Fresh α ′ → CEnv (Name α ′) α
→{m n : N}→ TG′ E α m n → TG′ _↼→_ α ′ m n

strengthenTG′ weak fr Γ (Output as) = Output (mapVarArgs (lookupCEnv Γ) as)
strengthenTG′ weak fr Γ (Let x f as u)

= let Γ′ = mapCEnv importWith Γ,weak x 7→ nameOf
in Let strongOf f (mapVarArgs (lookupCEnv Γ) as) (strengthenTG′ weak nextOf Γ′ u)

where open FreshPack fr

Both sequential and parallel composition are implemented by inserting the material of one graph between
the innermostLet and theOutput of the other graph. We define a general helper function for this purpose:

inLet′ : {α β : World}→ (s : α ∗
↼→ β )→ Fresh β →{m n n′ : N}

→ ({γ : World}→ (s′ : α ∗
↼→ γ)→ Fresh γ

→ Args γ m n → TG′ _↼→_ γ m n′)
→ TG′ _↼→_ β m n → TG′ _↼→_ β m n′

inLet′ s fr F (Let x f t u) = Let x f t (inLet′ (s � x) fr′ F u) where fr′ = StrongPack.nextOf x
inLet′ s fr F (Output as) = F s fr as

We first implement fork, which walks the only primitively available fresh linkfreshø past all theLets
of g1, uses the resulting fresh linkfr to renameg2, and afterwards adapts the output listas1 of g1 to the
inner world of the renamedg2, so that the two output lists can be concatenated:

forkTG′ : {m n1 n2 : N}
→ TG′ _↼→_ ø m n1

→ TG′ _↼→_ ø m n2

→ TG′ _↼→_ ø m (n1 + n2)
forkTG′ {m} {n1} {n2} g1 g2 = inLet′ ε freshø
(λ {γ} s′ fr as1 → inLet′ ε fr
(λ s′′ as2 → Output (mapVarArgs (import⊆ (∗↼→-⊆ s′′)) as1

+ as2))
(strengthenTG′ {_↼→_} StrongPack.weakOf fr emptyCEnv g2)

) g1

The implementation of parallel composition then relies on fork in the same way as that for TG:

parTG′ : {m1 n1 : N} → TG′ _↼→_ ø m1 n1

→ {m2 n2 : N}→ TG′ _↼→_ ø m2 n2

→ TG′ _↼→_ ø (m1 + m2) (n1 + n2)
parTG′ {m1} g1 {m2} g2 = forkTG′ (extendTG′ m2 g1) (shiftTG′ m1 g2)

Sequential composition follows the same pattern asforkTG′, and first traverses the declarations ofg1,
which are preserved, but uses the helper functionmapArgsTG′ to properly replace any occurrence of
inputs in argument and output lists of the renamedg2 with the corresponding output nodes ofg1, after
adapting them to the respective nested world.

seqTG′ : {k m n : N}
→ TG′ _↼→_ ø k m
→ TG′ _↼→_ ø m n
→ TG′ _↼→_ ø k n

seqTG′ g1 g2 = inLet′ ε freshø
(λ {γ} s′ fr as1 → mapArgsTG′ ε
(λ s′′ as → seqArgs (mapVarArgs (import⊆ (∗↼→-⊆ s′′)) as1) as)
(strengthenTG′ {_↼→_} StrongPack.weakOf fr emptyCEnv g2)

) g1

Finally, it is also reasonably easy to convert aTG’ term graph into aJungle with Fin k asInner node set
and asEdge set, wherek is the number ofLet declarations.

8 Conclusion and Outlook

Formalising mathematical definitions of term graphs and their operations in Agda is a remarkably straight-
forward exercise, and, due to the dependent typing of Agda, also carries over to typed term graphs much
more easily than in the more restricted type systems of Haskell or higher-order logic.

The remarkable abstract interface to variable binding provided by Pouillard and Pottier’sNotSoFresh

Agda library [PP10] also makes name-binding representations of term graphs conveniently accessible
to mechanised reasoning and programmed manipulation. Typing is easily added to ourTG andTG′

datatypes — the originalTm datatype provided asNotSoFresh example includes typing, but we omitted
it here to improve readability.

Implementing additional term graph operations, manipulations, and conversion functions, and prov-
ing the algebraic properties of the term graph operations isongoing work.

Future work will strive to base code-graph based optimised-code generation algorithms for the Co-
conut project [AK09a] on our Agda formalisations of code graphs, with a fully verifying tool chain as
ultimate goal.

